# $Id: PKGBUILD 169936 2012-10-30 22:45:47Z heftig $
# Maintainer: Ronald van Haren <ronald.archlinux.org>
# Contributor: Ben <ben@benmazer.net>

pkgname=gtkglext
pkgver=1.2.0
pkgrel=9
pkgdesc="opengl extensions for gtk2"
arch=('i686' 'x86_64')
url="http://gtkglext.sourceforge.net/"
license=('LGPL')
depends=('gtk2' 'glu' 'libxmu' 'pangox-compat')
makedepends=('mesa')
options=('!libtool')
source=("http://downloads.sourceforge.net/sourceforge/gtkglext/${pkgname}-${pkgver}.tar.bz2"
	'gtk2.20.patch')
md5sums=('ed7ba24ce06a8630c07f2d0ee5f04ab4'
         'e5a87ec3f2d0e616c6f32f90c3f7237f')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"

  patch -Np1 -i "${srcdir}/gtk2.20.patch"
  sed '/AC_PATH_XTRA/d' -i configure.in

  autoreconf --force --install
  ./configure --prefix=/usr \
              --disable-static
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
}
